



Stefanie Juchmes

(Jg. 1992) hat Informatik an der Universität Bonn studiert. Sie war schon auf einigen FileMaker Konferenzen und arbeitet seit Januar 2019 bei Monkeybread Software als Juniorentwicklerin.

stefanie@monkeybreadsoftware.de

[Link](#) MBS-Plugin | DynaPDF

Ordnung schaffen mit „DynaPDF“ und „MBS“

PDFs zusammenführen und Seitenzahlen einfügen

PDFs sind heutzutage aus der Geschäftswelt nicht mehr wegzudenken. Aufgrund seiner Plattformunabhängigkeit ist das Portable Document Format eines der am häufigsten gebrauchten Dateiformate. Das „MBS“-Plugin stellt Ihnen mit „DynaPDF“ viele Funktionen zur Verfügung, mit denen Sie Ihre PDF-Dateien bearbeiten und bereitstellen können. In diesem Artikel zeige ich Ihnen, wie Sie mehrere PDFs zu einer Datei zusammenführen und nachträglich Seitenzahlen einfügen.

Die meisten von Ihnen haben das sicher schon einmal so oder so ähnlich erlebt: Sie haben Ihre Rechnung als PDF vorliegen und möchten sie noch um Ihre AGB erweitern. Diese Erweiterung soll mit FileMaker über ein Script realisiert werden. Am Ende sollen in dem PDF noch Seitenzahlen eingefügt werden, damit man beim Ausdruck der Datei nicht durcheinanderkommt.

Zusammenfügen

Zunächst möchte ich Ihnen die Funktionalität vorstellen, die es ermöglicht, mehrere PDFs miteinander zu verbinden. Sie können ein vorhandenes PDF aus einem Containerfeld oder einer Datei importieren – die Ausgabe erfolgt ebenfalls in ein Containerfeld oder eine Datei (siehe Abbildung 1). Anschließend können Sie das PDF weiter verarbeiten, als PDF-Datei exportieren oder an einen Drucker senden (siehe FMM_201706, S. 25).



Nachfolgend finden Sie das Script zum Zusammenfügen von PDFs in Containern:

Run

```

◆ # DynaPDF initialisieren, falls nötig
◆ Wenn
  [ MBS("DynaPDF.IsInitialized") ≠ 1 ]
◆ Script ausführen
  [ Angegeben: Aus Liste ; "InitDynaPDF" ; Parameter: ]
◆ Ende (wenn)
◆ Wenn
  [ MBS("DynaPDF.IsInitialized") ≠ 1 ]
◆ Aktuelles Script verlassen
  [ Textergebnis: ]
◆ Ende (wenn)
◆ # Leeres neues PDF Dokument
◆ Variable setzen
  [ $pdf ; Wert: MBS("DynaPDF.New") ]
◆ # PDF aus dem Container laden
◆ Gehe zu Datens./Abfrage/Seite
  [ Erste(r) ]
◆ Variable setzen
  [ $destpage ; Wert: 1 ]
◆ Schleife (Anfang)
◆ Variable setzen
  [ $r ; Wert: MBS("DynaPDF.OpenPDFFromContainer"; $pdf; Merge PDFs::InputPDF) ]

```

```

◆ Variable setzen
[ $r ; Wert: MBS("DynaPDF.ImportPDFFile"; $pdf; $destpage) ]

◆ Wenn
[ LiesAlsZahl($r) > 0 ]

◆ Variable setzen
[ $destpage ; Wert: $r+1 ]

◆ Ende (wenn)

◆ Gehe zu Datens./Abfrage/Seite
[ Nächste(r) ; Nach letztem beenden: Ein ]

◆ Schleife (Ende)

◆ Feldwert setzen
[ Merge PDFs::FinalPDF ; MBS("DynaPDF.Save"; $pdf;
"MerGED.pdf") ]

◆ Variable setzen
[ $r ; Wert: MBS("DynaPDF.Release"; $pdf) ]
    
```

Als Erstes wird **DynaPDF** initialisiert, wofür Sie das Script „InitDynaPDF“ in Ihr Projekt kopieren und anpassen müssen. Anschließend wird mit „MBS (\"DynaPDF.New\")“ eine neue, leere PDF-Umgebung geschaffen.

Vom ersten Datensatz ausgehend läuft dann eine Schleife über alle Datensätze der aktuellen Tabelle bzw. des Suchergebnisses und öffnet für jeden Datensatz das im Containerfeld liegende PDF. Mit dem Aufruf der Funktion „MBS (\"DynaPDF.ImportPDFFile\"; \$pdf; \$destpage)“ werden alle PDF-Seiten in das neue PDF importiert¹. Im ersten Schleifendurchlauf hat die Variable **\$destpage** den Wert 1, sodass die PDF-Seiten an den Anfang unserer neuen Datei gestellt werden.

Anschließend wird die Seitenanzahl des erstellten Dokuments ermittelt und die Variable **\$destpage** erhält den Wert dieser Anzahl plus 1. Das bedeutet, dass beim nächsten Schleifendurchlauf das anzuhängende Dokument ab der nächsten freien Seite eingefügt wird. Anschließend springt das Script in den nächsten Datensatz und durchläuft die Schleife erneut.

Wenn die Schleife nach dem letzten Datensatz beendet wurde, wird das neue PDF in ein Containerfeld gespeichert. Optional kann dabei bereits ein Dateiname angegeben werden, der das Suffix **.pdf** enthalten sollte. Als weitere Option kann ein Vorschaubild erzeugt werden. Ist dies nicht gewünscht könnten Sie den Aufruf der **MBS**-Funktion wie folgt anpassen: „MBS (\"DynaPDF.Save\"; \$pdf; \"MerGED.pdf\"; 0)“. Wenn sie ein Vorschaubild erzeugen wollen, können sie die Null weglassen, so das der Standardwert 1 in der Funktion angenommen wird. Damit das Vorschaubild angezeigt wird, benötigen Sie eine **DynaPDF Pro**-Lizenz. Unter macOS können Sie alternativ die **PDFKit**-Bibliothek von Apple dazu verwenden.

Während das PDF in einem interaktiven Container über das Browserplugin dargestellt wird, braucht FileMaker in einem normalen Container für die Anzeige unter Windows oder in WebDirect ein Vorschaubild.

Im letzten Schritt wird die PDF-Umgebung mit „MBS (\"DynaPDF.Release\"; \$pdf)“ wieder freigegeben.

Nummerieren

Als Nächstes sollen einem PDF-Dokument Seitenzahlen hinzugefügt werden. Es empfiehlt sich, im Vorfeld bereits einen

Abb. 1 Mit dem **MBS**-Plugin können Sie mehrere PDFs zu einer Datei verbinden und z. B. eine Rechnung um die AGB erweitern.





Abb. 2 In diesem Beispiel sehen Sie, wie Sie Seitenzahlen in eine bereits vorhandene PDF-Datei einfügen. Die Ursprungsdatei wird dabei in das linke Containerfeld gezogen. Nach einem Klick auf den Button erscheint im rechten Container das Vorschaubild der PDF-Datei, die nun Seitenzahlen enthält und in dieses Containerfeld abgelegt wurde.



Bereich für die Seitenzahlen freizuhalten, sodass kein Text überschrieben wird (siehe Abbildung 2).

Nachfolgend finden Sie das Script für das Hinzufügen von Seitenzahlen:

Add Page Numbers

```

◆ # DynaPDF initialisieren, falls nötig
◆ Wenn
  [ MBS("DynaPDF.IsInitialized") ≠ 1 ]
◆ Script ausführen
  [ Angegeben: Aus Liste ; "InitDynaPDF" ; Parameter: ]
◆ Ende (wenn)
◆ Wenn
  [ MBS("DynaPDF.IsInitialized") ≠ 1 ]
◆ Aktuelles Script verlassen
  [ Textergebnis: ]
◆ Ende (wenn)
◆ # Leeres neues PDF Dokument
◆ Variable setzen
  [ $pdf ; Wert: MBS("DynaPDF.New") ]
◆ # Flags für Import
◆ Variable setzen
  [ $r ; Wert: MBS("DynaPDF.SetImportFlags";
  $pdf; "ImportAll ImportAsPage") ]
◆ #
◆ # Vorhandenes PDF importieren
◆ #
◆ Variable setzen
  [ $r ; Wert: MBS("DynaPDF.OpenPDFFromContainer"; $pdf;
  Add Page Numbers::Input) ]
◆ Variable setzen
  [ $r ; Wert: MBS("DynaPDF.ImportPDFFile"; $pdf; 1) ]
◆ #
◆ # Koordinatenursprung setzen
◆ Variable setzen
  [ $r ; Wert: MBS( "DynaPDF.SetPageCoords"; $PDF;
  "TopDown" ) ]
◆ # Seitennummern ergänzen
◆ Variable setzen
  [ $PageCount ; Wert: MBS( "DynaPDF.GetPageCount"; $pdf ) ]
◆ Variable setzen
  [ $PageNumber ; Wert: 1 ]
◆ Variable setzen
  [ $pageWidth ; Wert: MBS("DynaPDF.GetPageWidth"; $pdf) ]
◆ Variable setzen
  [ $pageHeight ; Wert: MBS("DynaPDF.GetPageHeight"; $pdf) ]
◆ Schleife (Anfang)
◆ Variable setzen
  [ $r ; Wert: MBS("DynaPDF.EditPage"; $pdf; $PageNumber) ]
◆ Variable setzen
  [ $r ; Wert: MBS( "DynaPDF.SetFont"; $pdf; "Helvetica";
  0; 20) ]
◆ Variable setzen
  [ $r ; Wert: MBS( "DynaPDF.WriteFTextEx"; $pdf; 50;
  $pageHeight - 50; $pageWidth-100; 30; "right";
  LiesAlsText($PageNumber) & " von " & $PageCount) ]
◆ Variable setzen
  [ $r ; Wert: MBS("DynaPDF.EndPage"; $pdf) ]
◆ Variable setzen
  [ $PageNumber ; Wert: $PageNumber +1 ]
◆ Verlasse Schleife wenn
  [ $PageNumber > $PageCount ]
◆ Schleife (Ende)
◆ # PDF sichern
◆ Variable setzen
  [ $PDFData ; Wert: MBS("DynaPDF.Save"; $pdf;
  LiesAlsText(Add Page Numbers::Input)) ]
◆ #

```

```

◆ # Daten freigeben
◆ #
◆ Variable setzen
  [ $r ; Wert: MBS("DynaPDF.Release"; $pdf) ]
◆ #
◆ # In den Container schreiben
◆ Feldwert setzen
  [ Add Page Numbers::Output ; $PDFData ]

```

Der Anfang dieses Scripts ist identisch mit dem vorherigen. Zunächst wird **DynaPDF** initialisiert und eine neue, leere PDF-Umgebung geschaffen, die in der Variablen **\$pdf** abgelegt wird. Anschließend werden einige Flags für den Import gesetzt, zum Beispiel das Flag „Import All“, das alle Seitenobjekte aus dem Ursprungsobjekt einfügt. Alternativ können auch nur bestimmte Objekte übernommen bzw. spezielle Seitenobjekte ausgelassen werden. „ImportAsPage“ sorgt dafür, dass die PDF-Seiten als Seiten und nicht als Vorlagen eingefügt werden. Danach wird die Datei mit dem schon bekannten Befehl importiert.

„MBS("DynaPDF.ImportPDFFile" ; \$pdf; 1)“ wurde oben bereits ausführlich erklärt. Kurz zur Erinnerung: Mit dieser Funktion wird ein PDF, das sich in einem Container befindet, in die PDF-Umgebung importiert. Der Aufruf „MBS("DynaPDF.SetPageCoords"; \$PDF; "TopDown")“ gibt die Koordinatenausrichtung vor, mit der gearbeitet werden soll. Ein PDF ist an einem Nullpunkt ausgerichtet, der sich normalerweise in der linken unteren Ecke einer Seite befindet (BottomUp). Für die meisten Menschen ist es aber angenehmer, wenn der Koordinatenursprung in der linken oberen Ecke liegt. Das bedeutet, je größer der Wert einer von uns angegebenen y-Koordinate ist, desto näher liegt der angegebene Punkt am unteren Seitenrand des PDF-Dokuments.

In **\$PageCount** wird die Seitenanzahl des zu nummerierenden Dokuments gespeichert. Zudem werden die Seitenhöhe und -breite in Variablen abgelegt. Anschließend wird eine Schleife gestartet, die das Dokument Seite für Seite durchgeht und die Seitenzahlen ergänzt. Über die Funktion „MBS("DynaPDF.EditPage"; \$pdf; \$PageNumber)“ wird die entsprechende Seite zum Bearbeiten geöffnet. Im nächsten Schritt wird für die richtige Formatierung gesorgt: Der Aufruf von „MBS("DynaPDF.SetFont"; \$pdf; "Helvetica"; 0; 20)“ setzt die Schriftart auf Helvetica, 20 pt. Die „0“ zwischen Schriftart und -größe steht für die Möglichkeit, einen Stil anzugeben, z. B. „1“ für Italic und „2“ für Fettschrift.

Mit „MBS("DynaPDF.WriteFTextEx"; \$pdf; 50; \$pageHeight - 50; \$pageWidth-100; 30; "right"; LiesAlsText(\$PageNumber) & " von " & \$PageCount)“ wird schließlich der Text in das Dokument geschrieben. Dafür wird zunächst die Position festgelegt, an der sich die Seitenzahl befinden soll – in unserem Fall am rechten unteren Rand. Der Parameter „right“ gibt die rechtsbündige Textausrichtung vor. Das letzte Argument betrifft den Text, der auf die Seite geschrieben werden soll. Wie Sie hier sehen, kann man dem PDF nicht nur Seitenzahlen hinzufügen, sondern es können beliebige Texte an jeder gewünschten Position integriert werden.

Abb. 3 Nicht nur das Einfügen von Seitenzahlen ist möglich, sondern auch das Hinzufügen von Texten an beliebiger Position



Im Anschluss wird die Seite geschlossen und die aktuelle Seitenzahl um eins hochgezählt. Die Schleife durchläuft die restlichen Seiten bis zum Ende des Dokuments, wo sie beendet und das fertige PDF in einem Containerfeld gespeichert wird.

Zwei in einem

Wenn Sie mehrere PDFs zu einer Datei zusammenfügen und diese nummerieren möchten, führen Sie zunächst das erste Script aus. Vor dem Speichern und Freigeben des Dokuments können Sie das zweite Script ab der Festlegung der Koordinaten einfügen. Denken Sie bitte daran, zuvor die Flags zu setzen.

Lizenzen

Um mit den vorgestellten Funktionen arbeiten zu können, benötigen Sie Lizenz(en) des **MBS-Plugins** und eine **DynaPDF-Lite**-Lizenz (oder höher). Wenn Sie einem PDF einzelne Seiten hinzufügen oder Vorschaubilder erzeugen möchten, ist eine **DynaPDF-Pro**-Lizenz notwendig. Diese Lizenzen erhalten Sie auf unserer Webseite, wo Sie auch weitere Informationen finden. Sie können das **MBS-Plugin** und die **DynaPDF**-Lizenz als Probeversionen testen.

„DynaPDF“ initialisieren

Bevor Sie mit **DynaPDF** arbeiten, muss es einmal initialisiert werden, dies übernimmt das Script „InitDynaPDF“. Legen Sie die Datei *dynapdf.dylib* (Mac) bzw. *dynapdf.dll* (Windows) entweder auf Ihrem Server oder in einem Ordner auf Ihrem Client ab und passen Sie den Pfad zu dieser Datei in „InitDynaPDF“ an. Voreingestellt ist der Ordner, in dem Ihre Datenbank liegt, aber Sie können natürlich Ihren eigenen Pfad einfügen. Wenn Sie lediglich den Dateinamen angeben, sucht das Script in dem Ordner, in dem Sie das **MBS-Plugin** gespeichert haben. *dynapdf.dylib*, *dynapdf.dll*, sowie alle Beispieldateien finden Sie in dem Examples-Ordner, den Sie mit dem **MBS-Plugin** downloaden.

Ich wünsche Ihnen viel Spaß beim Ausprobieren! ■

Fußnote

- 1) Wenn Sie nur einzelne Seiten an Ihr PDF-Dokument anfügen wollen, steht Ihnen die Funktion „DynaPDF.ImportPage“ zur Verfügung, die auch mehrfach hintereinander verwendet werden kann

FileMaker Magazin



Das FileMaker Magazin

- Einzige, deutschsprachige Fachzeitschrift zu FileMaker
- Wissen aus erster Hand von anerkannten FileMaker Fachautoren
- Große Themenvielfalt für Anwender und Entwickler

Exklusiv für Premium-Abonnenten

- Sechs FMM Ausgaben pro Jahr
- Kostenlose Nutzung des Abonnentenbereichs auf www.filemaker-magazin.de
- PDF-Archiv mit allen bisher veröffentlichten Ausgaben
- Jede Ausgabe mit kostenlosen Beispieldateien und Zusatzinfos zum Download

Unser Service

- Aktuelle Neuheiten, Tipps und Infos, Kleinanzeigen und vieles mehr jederzeit auf unseren Webseiten
- Hilfe bei allen Fragen zu FileMaker im FMM Forum
- Kompetente Beratung zum Kauf von FileMaker Lizenzen: Einfach anrufen +49 (0)40 589 65 79 70.

Wenn Sie sich für ein FileMaker Magazin
Abo interessieren, klicken Sie bitte hier!

Eine kostenlose **Leseprobe** des FileMaker Magazins erhalten Sie, wenn Sie hier klicken.

Hier finden Sie Aktuelles zu **FileMaker Lizenzen**, egal ob Sie kaufen, mieten oder sich einfach informieren möchten.